
RTC 72421 auslesen 
 
 

 

Die GIDE+USB-Platine enthält mit dem bei Bedarf zu 
bestückenden 72421 auch Echtzeituhr und Kalender. Nachfolgend 
werden dazu einige Hinweise incl. lauffähiger Beispiele gegeben.  
 
Sie stellen lediglich die nötigen Angaben für das Grundverständnis 
dar und können als Anregung für eigene Experimente dienen. So 
wird z.B. auch auf das doppelte Auslesen der Zeit (um Probleme 
bei Änderungen der Zeit im Auslesemoment zu vermeiden) 
verzichtet.  
 
Nicht eingegangen wird hier auf die Programmierung (Stellen der 
Uhr). Das kann z.B. bequem mit dem Programm GIDE-Check von 
Ralph Hänsel erfolgen.  

 
 

 
Grundlagen 
 
Für das Ansprechen der Echtzeituhr ist eine 16-Bit-Adresse nötig: 
 
Register- 
adresse Bedeutung 

0085 Sekunden, Einer 
0185 Sekunden, Zehner 
0285 Minuten, Einer 
0385 Minuten, Zehner 
0485 Stunden, Einer 
0585 Stunden, Zehner & AM/PM flag 
0685 Tag, Einer 
0785 Tag, Zehner 
0885 Monat, Einer 
0985 Monat, Zehner 
0A85 Jahr, Einer 
0B85 Jahr, Zehner 
0C85 Wochentag (0=Sonntag, 1=Montag,...) 
0D85 Steuerregister D  (status/control) 
0E85 Steuerregister E  (pulse output control) 
0F85 Steuerregister F  (master control) 

 
 
Der niederwertige Teil 85h steht hier für den Uhrenport, was sich aus der u.a. beim LLC2 
angewandten Basis-Adressierung des GIDE+USB-Moduls (80h) ergibt.  Der höherwertige Teil der 
Adresse identifiziert eines der insgesamt 16 Register. 
 
Einen einfachen 16-Bit-Lese-Zugriff realisiert automatisch z.B. der Befehl IN A,(C):  
 
Mit  85h   in Register C (Daten liegen beim Lesen an A0...A7) 
und 00...0Fh in Register B (Daten liegen beim Lesen an A8...A15) 
 
holt IN A,(C) den Wert aus dem entsprechenden Register nach A. Da die oberen 4 Bit des 
Registerinhalts nicht relevant sind, führt man anschließend noch AND 0Fh aus. Das Ergebnis ist eine 
Zahl von 0...15 und ist ggf. noch zu konvertieren. 

Beispiele: 
 
MC-Programm zum Lesen von Datum und Uhrzeit (RTC_1900.BIN) 
 
PUFFER:   EQU 1880H 
RTCPORT:  EQU 85H 
ORG 1900H 
       DEFM 0,9,'z',0DH  ;VORSPANN, KENNBUCHSTABE 'z' 
       LD HL,PUFFER      ;PUFFERADRESSE 
       PUSH HL           ;ANFANG SICHERN 
       LD (HL),0DFH      ;RST18 FÜR AUSGABE ABLEGEN 
       INC HL 
       LD C,RTCPORT      ;IO-Adresse RTC 
       LD B,0CH          ;ZU LESENDES RTC-REGISTER (START) 
       LD D,6            ;6 x 2 STELLEN 
LESEN: DEC B             ;ALLE GEWÜNSCHTEN REGISTER ABKLAPPERN 
       IN A,(C)          ;RTC BYTE1 HOLEN (B=REGISTER!) 
       AND 0FH           ;NUR NIEDERWERTIGER TEIL 
       OR 30H            ;ALS ASCII-WERT 
       LD (HL),A         ;ABLEGEN 
       INC HL 
       DEC B 
       IN A,(C)          ;RTC BYTE2 HOLEN 
       AND 0FH 
       OR 30H             
       LD (HL),A          
       INC HL 
       LD (HL),'/'       ;TRENNER FÜR DATUM 
       INC HL 
       DEC D 
       JR NZ,LESEN 
       POP IX            ;PUFFERANFANG=STARTADRESSE 
       LD (IX+09H),' ' 
       LD (IX+0CH),':'   ;NEUE TRENNER FÜR ZEIT  
       LD (IX+0FH),':' 
       LD (IX+12H),0A0H  ;ENDE FÜR RST18-AUSGABE 
       LD (IX+13H),0C9H  ;RÜCKKEHR 
      ;AUSGABEART--------------------------------- 
       LD A,(185BH)      ;ARGUMENT HOLEN 
       CP 3              ;>=3? 
       RET NC            ;JA, NUR PUFFER BEFÜLLT 
       CP 2              ;2? 
       JR Z,KOMPL        ;JA = KOMPLETTAUSGABE 
       CP 0              ;0 ODER KEIN ARGUMENT?  
       JR Z,ZEIT         ;JA = NUR ZEIT 
DAT:   SET 7,(IX+09H)    ;NUR DATUM AUSGEBEN 
       LD (IX+0AH),0C9H  ;VORZEITIGES RET 
       JR KOMPL 
ZEIT:  LD (IX+09H),0DFH  ;VERZÖGERTES RST18 
       LD IX,PUFFER+9    ;NEUER ANFANG AUSGABE 
KOMPL: JP (IX)           ;ZK AUSGEBEN UND ZURÜCK ZUM AUFRUFENDEN PROGRAMM 
Das Programm gibt (wenn Option <>3) den Puffer als ASCII an der momentan 
aktuellen BS-Position aus. 
Aufruf aus Monitor: 
z 0   nur Uhrzeit (0 kann entfallen) 
z 1   nur Datum  
z 2   Datum+Zeit 
z 3   keine Anzeige, nur Puffer befüllen 

Aufruf aus Anwenderprogramm, z.B.: 
 
LD   A,1 
CALL 1904 
 

PUFFERSTRUKTUR/INHALT ab 1880h: 
0  1  2  3  4  5  6  7  8  9  A  B  C  D  E  F  10 11 12 13 
DF J  J  /  M  M  /  D  D     H  H  :  M  M  :  S  S     C9 
   10 1     10 1     10 1     10 1     10 1     10 1 

Unter GS-BASIC ist mit der INP()-Funktion zwar ein direkter Portzugriff möglich, aber nicht in 16 Bit 1. 
Es kann nur die Uhrenportadresse (85h) übergeben (gelesen) werden. Der höherwertige Teil der 
Adresse ist immer 00, womit lediglich die Sekunden/Einer gelesen werden. Der Zugriff auf die 
anderen Daten sowie die Steuerregister ist damit verwehrt. INP(133) holt also nur die 
Sekunden/Einer. Mit X=INP(133) AND 15 werden die nicht relevanten oberen 4 Bits ausgeblendet, 
sodass sich ein Wert von 0...15 ergibt. Das kann als Zeitgeber in einigen Fällen bereits genügen. 
  
Für die Erlangung von aktueller Zeit sowie Datum ist etwas MC-Code zu benutzen. Er läuft hier im 
Bereich 1900h...190Bh. Für einen anderen Bereich ist der Code entsprechend anzupassen. Als 
Übergabezelle für das gewünschte Uhrenregister ist 1901h definiert.  
 
MC-Teil: 
 
19ØØ Ø6 ØC  LD B,ØCH       ;UHRENREGISTER-NR. IN 19Ø1 
19Ø2 ØE 85  LD C,85H  ;UHRENPORT 
19Ø4 ED 78  IN A,(C)     ;RTC BYTE HOLEN 
19Ø6 E6 ØF  AND ØFH  ;HWT AUSBLENDEN 
19Ø8 32 Ø1 19 LD (19Ø1),A  ;RÜCKGABEWERT IN 19Ø1 
19ØB C9  RET 
 
BASIC-Teil: 
Der MC-Code kann z.B. mit der POKE-Methode zusammen mit dem BASIC-Progamm abgespeichert 
werden. 
 
Vollständiges Beispiel in GS-BASIC (RTCBAS.z80) 
 
 1Ø  CLS:PRINT"RTC-DATUM UND UHRZEIT AUSLESEN:" 
 2Ø  GOSUB24Ø     :REM MC poken 
 3Ø  F=7:GOSUB21Ø     :REM Tag 10er 
 4Ø  F=6:GOSUB21Ø     :REM Tag 1er 
 5Ø  PRINT"."; 
 6Ø  F=9:GOSUB21Ø     :REM Monat 10er 
 7Ø  F=8:GOSUB21Ø     :REM Monat 1er 
 8Ø  PRINT"."; 
 9Ø  F=11:GOSUB21Ø    :REM Jahr 10er 
1ØØ  F=1Ø:GOSUB21Ø    :REM Jahr 1er  
11Ø  PRINT", "; 
12Ø  F=5:GOSUB21Ø     :REM Stunden 10er 
13Ø  F=4:GOSUB21Ø     :REM Stunden 1er 
14Ø  PRINT":"; 
15Ø  F=3:GOSUB21Ø     :REM Minuten 10er 
16Ø  F=2:GOSUB21Ø     :REM Minuten 1er 
17Ø  PRINT":"; 
18Ø  F=1:GOSUB21Ø     :REM Sekunden 10er 
19Ø  F=Ø:GOSUB21Ø     :REM Sekunden 1er 
2ØØ  END 
21Ø  POKE64Ø1,F:CALL*19ØØ:X=PEEK(64Ø1) :REM RTC lesen 
22Ø  X=X+48:PRINTCHR$(X);   :REM als ASCII ausgeben 
23Ø  RETURN 
24Ø  RESTORE27Ø     :REM MC-Code 
25Ø  FORI=ØTO11:READX:POKE64ØØ+I,X:NEXT :REM poken 
26Ø  RETURN 
27Ø  DATA 6,12,14,133,237,12Ø,23Ø,15,5Ø,1,25,2Ø1   :REM obiger MC-Code dezimal 
 
 
Erstellt:  
WeRo 
Stand: Nov. 2013 
 

                                                
1 Zumindest ist mir aus den verfügbaren Unterlagen keine Möglichkeit bekannt. 


